

# Initiation à MAPLE

## § 1. Introduction

▲ Maple est un logiciel (un système) de calcul très performant. Il effectue les calculs à la fois numériques et formels.

**Calcul formel** signifie : **calcul symbolique**, calcul sur des lettres. Il signifie également : **calcul exact**. On demande par exemple à Maple de résoudre l'équation trinômiale

$$tx^2 + 3tx + 1 = 0,$$

où x est l'inconnue et t, un paramètre ; Maple répond :

$$\frac{1}{2} \frac{-3t + \sqrt{9t^2 - 4t}}{t}, \frac{1}{2} \frac{-3t - \sqrt{9t^2 - 4t}}{t}$$

Maple fonctionne *en mode direct (interactif)*, à l'image d'une calculatrice très puissante, et *en mode programmation*. Il nous offre l'opportunité de créer nos propres programmes, procédures, modules, ... Les procédures confectionnées seront utilisées au même titre que les fonctions Maple (prédéfinies). Voir des exemples de procédures, au § 4.

▲ Toute fenêtre ouverte sous Maple s'appelle **une feuille de calcul (worksheet)**, en anglais.) Le prompt de Maple est, par défaut, le symbole **>** après lequel toute saisie doit être effectuée. Par exemple, pour calculer  $\sqrt{13}$ , on tape puis on valide en appuyant sur la touche **Entrée** :

**> sqrt(13.) ;**

N'oublions pas le point-virgule ! sinon, Maple retourne le message suivant :

Warning, premature end of input

signifiant : **attention ! fin de commande précipitée.**

**Après validation**, Maple affiche le résultat suivant

3.605551276

Notons que le point placé après le nombre 13, signifie que ce nombre est considéré comme nombre réel. En l'absence de ce point, Maple aurait répondu

$$\sqrt{13}$$

En effet, la valeur obtenue précédemment pour  $\sqrt{13}$ , n'est qu'approximative. Rappelons nous ;  $\sqrt{13}$  est un nombre irrationnel. Tant que possible, Maple garde la valeur exacte du résultat. Voici un autre exemple permettant de mieux comprendre.

**> 7+8/13 ;**

$$\frac{99}{13}$$

## Initiation à Maple (1<sup>ère</sup> partie)

par Ghouti BOUKLI-HACENE

E-mail: [gh.boukli@gmail.com](mailto:gh.boukli@gmail.com) [g\\_boukli@yahoo.com](mailto:g_boukli@yahoo.com)

---

Maple présente le résultat sous forme fractionnaire (**résultat exact**). On peut le forcer à donner une réponse en nombres réels, en mettant quelque part, un point flottant.

```
> 7.+8/13 ;
```

7.615384615

Un autre moyen pour forcer Maple à fournir cette réponse, est :

```
> evalf(7+8/13) ;
```

Cette commande signifie : évaluer en tant que réel (**evaluate as a float**).

La commande à saisir pour résoudre l'équation citée plus haut, est

```
> solve(t*x^2+3*t*x+1=0,x);
```

▲ Maple fait les distinctions entre majuscules et minuscules. Par exemple, la ligne de commande

```
> sin(Pi/2);
```

fournit, après validation, le résultat

1

alors que

```
> Sin(Pi/2);
```

fournit

$\text{Sin}\left(\frac{1}{2}\pi\right)$

On appelle cela **la forme inerte** de la commande précédente (qualifiée elle, de **forme active**.)

**Un autre exemple :**

```
> Sum(k**4,k=1..n) = sum(k**4,k=1..n);
```

$$\sum_{k=1}^n k^4 = \frac{1}{5}(n+1)^5 - \frac{1}{2}(n+1)^4 + \frac{1}{3}(n+1)^3 - \frac{1}{30}n - \frac{1}{30}$$

Pour effectuer le produit, il suffit de remplacer **sum** par **product**. Par exemple,

```
> product(i,i=5..11);
```

1663200

Le même résultat est obtenu en validant

```
> 5*6*7*8*9*10*11;
```

▲ **Retour au prompt.** Si le symbole > ne vous plaît pas, vous pouvez le modifier à votre convenance. Tenez ! tapez la commande suivante

```
> interface(echo=2, prompt="coucou, bonjour!");
```

puis validez et observez le résultat. Pour faire apparaître ce prompt à un endroit donné de la feuille de calcul, il suffit d'y placer le curseur puis de cliquer sur le bouton .

▲ Pour obtenir les lettres grecques, il suffit de taper leurs noms comme suit :

<i>Lettre</i>	<i>Saisie</i>	<i>Lettre</i>	<i>Saisie</i>
α	alpha	A	Alpha
β	beta	B	Beta
γ	gamma	Γ	Gamma
δ	delta	Δ	Delta
ε	epsilon	E	Epsilon
ζ	zeta	Z	ZETA
η	eta	H	Eta
θ	theta	Θ	Theta
ι	iota	I	Iota
κ	kappa	K	Kappa
λ	lambda	Λ	Lambda
μ	mu	M	Mu
ν	nu	N	Nu
ξ	xi	Ξ	Xi
ο	omicron	O	Omicron
π	pi	Π	PI
ρ	rho	P	Rho
σ	sigma	Σ	Sigma
τ	tau	T	Tau
υ	upsilon	Υ	Upsilon
φ	phi	Φ	Phi
χ	chi	X	CHI
ψ	psi	Ψ	Psi
ω	omega	Ω	Omega

▲ Gardons en mémoire :

> `evalf(Pi);`

3.141592654

La commande `evalf` permet d'évaluer le nombre indiqué en tant que nombre réel, avec **dix chiffres significatifs, par défaut**. On peut diminuer ou augmenter ce nombre à notre convenance. La ligne de commande suivante, fournit une valeur approchée de la constante  $\gamma$  d'Euler, avec 58 chiffres significatifs.

```
> evalf(gamma,58) ;  
      .5772156649015328606065120900824024310421593359399235988058
```

Si, dans une feuille de calcul Maple, on désire effectuer tout un ensemble de calculs avec 13 chiffres significatifs, par exemple, il est préférable de commencer par saisir et valider la commande suivante.

```
> Digits := 13 ;
```

Voici un exemple :

```
> Digits:=3 : 2.17*19 ; evalf(Pi) ; 5/17. ;  
      41.2  
      3.14  
      .294
```

▲ Maple dispose de palettes (de symboles, expressions, ...). Cliquer sur « View » dans la barre de menus, puis sur « Palettes » ...

▲ Dans une feuille de calcul Maple, le caractère % indique le **résultat précédent** (en mémoire.) Notons que tous les résultats obtenus sont mémorisés selon l'ordre chronologique ; peut importe leurs endroits d'apparition dans la feuille de calcul. Deux fois le caractère % indique l'**avant dernier résultat**, et trois fois le caractère % indique l'**antépénultième résultat** (c'est-à-dire, trois résultats en arrière). Voici un exemple :

```
> 1.5*3; (%+1)/2; %%+4; %%%*2;  
      4.5  
      2.750000000  
      8.5  
      9.0
```

▲ Comme on l'a dit plus haut, dans une feuille de calcul toutes les actions, tous les résultats obtenus sont mémorisés par Maple. On peut d'ailleurs enregistrer la feuille de calcul, la fermer, puis l'ouvrir ultérieurement et continuer le travail à l'endroit où s'est arrêté, modulo quelques précautions.

Supposons qu'à un moment donné, on a affecté à une variable identifiée par a, une valeur (disons, numérique) donnée, alors cette affectation reste en vigueur, dans toute la feuille de calcul, lors de la session en cours. On peut « désaffecter » cette variable, par un **unassign**. Regardez :

```
> a:=13: print(a);  
      13  
  
> unassign('a'): print(a);  
      a
```

A la suite d'un **restart**, toutes les variables sont désaffectés. Il s'agit d'une ré-initialisation de la feuille de calcul.

▲ *Il y a deux points et deux points ...*

: ( lire : **deux points** ) à la fin d'une ligne de commande, permet d'exécuter cette dernière, sans afficher quoi que ce soit à l'écran.

**:=** indique l'affectation, l'assignation.

**:-** permet de repérer les noms (ou membres) exportés d'un module (voir un exemple d'application dans § 7.)

**..** ( lire : **point point** ) indique un intervalle, comme dans

```
> sum(k**4, k=1..n);
```

**::** ( lire : **deux points deux points** ) indique une éventuelle déclaration de type dans une procédure. Voir § 4.

▲ Lors de la saisie, pour passer à la ligne suivante, il suffit d'appuyer simultanément sur les touches **Maj** et **Entrée**.

## § 2. Opérateurs et fonctions prédéfinies

▲ **Opérateurs arithmétiques :**

+ pour l'addition,

- pour la soustraction,

\* pour la multiplication,

/ pour la division (réelle),

^ ou \*\* pour l'élévation à une puissance,

&\* pour une multiplication non commutative (la multiplication de matrices par exemple.)

▲ **Opérateurs d'ordre :**

< (inférieur à)

<= (inférieur ou égal à)

> (supérieur à)

>= (supérieur ou égal à)

= (égal à)

<> (différent de)

▲ **Opérateurs logiques :** not and or

▲ L'opérateur dollar \$ indique une séquence. **Exemple :**

```
> k $ k=1..7;
```

1, 2, 3, 4, 5, 6, 7

**Un autre exemple :** la ligne de commande

```
> diff(x**3*y**7+sin(x**2+y+exp(2*z)), x$2,y$3,z) ;
```

permet (après validation) de calculer la dérivée partielle mixte

$$\frac{\partial^6}{\partial z \partial y^3 \partial x^2} (x^3 y^7 + \sin(x^2 + y + e^{(2z)}))$$

▲ -> (flèche) est utilisée pour définir une fonction. **Exemple :**

```
> f:=x->x**2+cos(x) ;
```

On peut s'en servir pour saisir une suite réelle :

```
> u:=n->exp(n)/(n+1); u(0); u(3), evalf(u(3));
```

$$u := n \rightarrow \frac{e^n}{n+1}$$
$$\frac{1}{4} e^3, 5.021384230$$

▲ @ (arobase) est l'opérateur de composition des fonctions.

```
> f:=t->2*t+1; (f@f)(x); (f@5)(x);
```

$$f := t \rightarrow 2t + 1$$
$$4x + 3 \quad (f \text{ composée } f)$$
$$32x + 31 \quad (f \text{ composée } f \text{ 5 fois})$$

▲ Les accolades sont **des constructeurs d'ensembles**, comme sur le papier ; **voici un exemple**

```
> A:={1,2,3,5} : B:={2,3,8,11,12} :  
C1:=A union B ; C2:=A intersect B ; C3:=A minus B ;
```

$$C1 := \{1, 2, 3, 5, 8, 11, 12\}$$
$$C2 := \{2, 3\}$$
$$C3 := \{1, 5\}$$

▲ Les crochets sont **des constructeurs de listes**. Par exemple, pour construire une matrice à 3 lignes et 4 colonnes, on saisit (et on valide) la commande

```
> matrix(3,4,[ ]);
```

où l'on saisit à l'intérieur des crochets les douze éléments de la matrice, séparés par des virgules. Maple prend alors le soin de les répartir, de gauche à droite, suivant les lignes de la matrice. On y reviendra.

▲ Outre les séquences, ensembles et listes, Maple dispose de deux autres structures : **les tables** et **les tableaux**. Nous en reparlerons.

▲ Comme **gamma** et **Pi**, la lettre capitale **I** est un objet Maple protégé, qui dénote le nombre complexe de carré  $-1$ . Il existe deux manières de saisir un nombre complexe sous Maple :

### forme algébrique

$$a + ib, (\mathbf{a+I*b}, \text{ sous Maple})$$

où  $a$  est la partie réelle du nombre complexe et  $b$ , sa partie imaginaire. Nous verrons plus bas comment les isoler.

### forme polaire

$$re^{it}, (\mathbf{polar(r,t)}, \text{ sous Maple}),$$

où  $r$  est le module du nombre complexe et  $t$ , son argument.

Pour passer de la forme polaire à la forme algébrique, on utilise la fonction **evalc**, ( **evaluate as a complex** ), et pour passer de la forme algébrique à la forme polaire, on fait usage de la fonction de conversion **convert**.

```
> z1:=2+3*I; z2:=-1+2*I;
```

$$z1 := 2 + 3 I$$

$$z2 := -1 + 2 I$$

```
> z1+z2, z1*z2, z1/z2;
```

$$1 + 5 I, -8 + I, \frac{4}{5} - \frac{7}{5} I$$

```
> Z:=polar(2,Pi/5);
```

$$Z := \text{polar}\left(2, \frac{1}{5} \pi\right)$$

```
> convert(z1,polar);
```

$$\text{polar}\left(\sqrt{13}, \arctan\left(\frac{3}{2}\right)\right)$$

```
> evalc(Z);
```

$$2 \cos\left(\frac{1}{5} \pi\right) + 2 I \sin\left(\frac{1}{5} \pi\right)$$

```
> Digits:=3: evalf(evalc(Z));
```

$$1.62 + 1.18 I$$

▲ **infinity** indique plus infini sous Maple.

```
> infinity ;
```

$\infty$

▲ Le dièse # permet d'insérer des commentaires, dans une ligne de commande. **Exemple :**

```
> evalf(Pi,10000); #Ceci permet d'évaluer la constante Pi avec 10000 chiffres.
```

▲ **Quelques fonctions Maple**

*Fonctions circulaires et leurs inverses :*

**sin(x), cos(x), tan(x)**  
**arcsin(x), arccos(x), arctan(x)**

*Fonctions hyperboliques et leurs inverses :*

**sinh(x), cosh(x), tanh(x),**  
**arcsinh(x), arccosh(x), arctanh(x)**

*Autres fonctions :*

Pour x, réel,

**ln(x)** ou **log(x)** (logarithme népérien de x),  
**exp(x)** (exponentielle naturelle de x),  
**log[10](x)** (logarithme décimal de x)  
**sqrt(x)** (racine carrée de x),  
**surd(x,n)** (racine nième de x)  
**abs(x)** (valeur absolue de x)  
**floor(x)** (partie entière de x)

Etant donnés deux entiers relatifs a et b,

**iquo(a,b)** (quotient de la division euclidienne de a par b)  
**irem(a,b)** (reste de la division euclidienne de a par b)

Etant donnés deux polynômes A et B en x,

**quo(A,B,x)** (quotient de la division euclidienne de A par B)  
**rem(A,B,x)** (reste de la division euclidienne de A par B)

Pour n, k, entiers naturels,

**factorial(n)** (n factoriel)  
**binomial(n,k)** (coefficient binomial  $C_n^k$ )

Pour z, complexe,

**Re(z)** (partie réelle de z)  
**Im(z)** (partie imaginaire de z)  
**abs(z)** (module de z)  
**argument(z)** (argument de z)  
**conjugate(z)** (nombre conjugué de z)